Planeten Bewegungen

Dr. Wilfried Tenten

31.03.2007

Dieses kleine MAPLE Beispiel soll mit einfachen Gleichungen die Planeten- und Kometenbahnen aus Erdsicht zeigen. Exemplarisch
wurde dazu der Planet Merkur und der Komet Halley ausgesucht. Andere Planeten, Asteroiden oder Kometen lassen sich durch Einsetzen Ihrer Ephemeriden leicht einfügen.

Es wird zunächst mit einer reinen Ellipsenbahn gerechnet und anschließend die Kepplerbbahnen betrachtet. Relativistische Effekte sowie Bahnstörungen durch Monde und andere Himmelskörper bleiben hier unberücksichtigt.
Das soll ja auch nur ein einfaches Beispiel sein, das sich mit dem Anspruch eines Schülerwissens der gymnasialen Oberstufe leicht verstehen lassen soll.

Die Ephemeriden sind von: http://www.solarviews.com/germ/data1.html und http://www.solarviews.com/germ/data2.html

```
> restart;
```

Kreis- und Elliptische Bewegungen

Erst mal die Kreisbewegung mit Radius r und der Rotationsperiode omega:
x(t) = a*cos(2*Pi*t/omega), y(t) = a*sin(2*Pi*t/omega).

Die Erdbahn besitzt einen mittleren Radius von 6378.14 km, omega = 1 Jahr.

```
> r_erde := 6378.14:  omega_erde := 1:
x_erde := t -> r_erde*cos(2*Pi*t/omega_erde):  y_erde :=
t ->r_erde*sin(2*Pi*t/omega_erde):
plot([x_erde(t), y_erde(t), t=0..1],title="Erdbahn als
Kreisbahn" );
```



Jetzt erweitern wir diese Sicht auf die Ellipsenbahn. Die Exzentrizität beträgt: 0.0167.
Zeichnen wir das mal, und wir werden sehen: aus dem schönen Rund wird ein Ei.

```
> e_erde := .0167:
c_erde := r_erde*e_erde;
b_erde := r_erde*sqrt(1-e_erde^2);
```

$$c\_erde := 106.514938$$

$$b\_erde := 6377.250538$$

Die Ellipsen verlangen eine etwas andere mathematische Darstellung, denn der Kreisradius wird einerseits gestreckt
andererseits gestaucht. Genannt wird dies dann die große bzw. die kleine Halbachse:

```
> x_erde := t -> r_erde*cos(2*Pi*t/omega_erde) + c_erde:
y_erde := t -> b_erde*sin(2*Pi*t/omega_erde):
plot([x_erde(t), y_erde(t), t=0..1], title="Erdbahn mit
ihrer Exzentizität" );
```



Schauen wir und die sonnennächste Bahn an: die des Merkur:

```
> r_merk := 2439.7:   omega_merk := evalf((r_merk/r_erde)^(3/2)):
e_merk := .2056:    c_merk := r_merk*e_merk:    b_merk := r_merk*sqrt(1-e_merk^2):
x_merk := t -> r_merk*cos(2*Pi*t/omega_merk) + c_merk:
y_merk := t -> b_merk*sin(2*Pi*t/omega_merk):
plot([x_merk(t), y_merk(t), t=0..omega_merk], scaling=constrained);
```



Vergleich mit der idealen Kreisbahn des Merkur:

```
> xc_merk := t -> r_merk*cos(2*Pi*t/omega_merk) + c_merk:
```

```
yc_merk := t -> r_merk*sin(2*Pi*t/omega_merk):
plot({[xc_merk(t),yc_merk(t), t=0..omega_merk], [x_merk
(t), y_merk(t), t=0..omega_merk]},
scaling=constrained,color=[red,blue],title="Vergleich der
Merkurbahn: Kreisbahn (rot) mit Ellipsenbahn (blau)");
```

Vergleich der Merkurbahn: Kreisbahn (rot) mit Ellipsenbahn (blau)



Vergleichen wir diese Bahnen mit dem berühmten Halley-Komet

```
> r_halley := 115216:   omega_halley := evalf
((r_halley/r_erde)^(3/2));
e_halley := .97:    c_halley := r_halley*e_halley:
b_halley := r_halley*sqrt(1-e_halley^2):
x_halley := t -> r_halley*cos(2*Pi*t/omega_halley) +
c_halley:
y_halley := t -> b_halley*sin(2*Pi*t/omega_halley):
plot([x_halley(t), y_halley(t), t=0..omega_halley],
scaling=constrained);
```

$$omega\_halley := 76.77646615$$



Der Halley Komet benötigt 77 Jahre für einen Sonnenumlauf und seine Ellipse ist sehr gestreckt.

Relative Bewegungsung und deren Bahnfiguren

Oben habe ich die Bahnen der Planeten vorgestellt. Das ist zwar unter Vernachlässigung aller Nebeneffekte wie Monde, Beeinflussung anderer Planeten passiert
und auch relativistische Effekte sind nicht vorhanden, also keine Periheldrehungen. Darauf kommt es auch hier nicht an.
Schauen wir jetzt mal auf die relative Bewegung der Planeten zueinander. Wir, die Menschen auf der Erde, schauen den Merkur an. Wie sehen wir seine Bahn?
Der Merkur als auch die Erde sind in ständiger Bewegung und der Merkur ist sonnennäher, bewegt sich demnach schneller.

Also

```
> tmax := max(2*omega_merk, 2*omega_erde):
plot([x_merk(t) - x_erde(t), y_merk(t) - y_erde(t), t=
0..tmax], scaling=constrained,
> title="Die Merkur Bahn von der Erde aus gesehen (Erde im
Mittelpunkt)");
```



Wir beobachten eine seltsame wellenförmige Bewegung. Diese Wellenform hat offensichtlich mit der Differenzbildung der Einzelbewegungen zu tun.
Schauen wir in den Himmel und beobachten den Mekur, so sehen wir, daß er sich auf- und ab bewegt, daß er sogar stehenbleibt und sich gar
rückwärts bewegt. Vergleicht das mal diesem Ergebnis und schaut mal, wie einfach das Ergebnis zu erhalten war!!!

Wagen wir uns damit mal an Halley:

```
> tmax := max(2*omega_halley, 2*omega_erde):
plot([x_halley(t) - x_erde(t), y_halley(t) - y_erde(t),
t=0..tmax], scaling=constrained,
> title="Die Kometenbahn des Halley von der Erde aus
gesehen, Erde im Mittelpunkt)");
```

Die Kometenbahn des Halley von der Erde aus gesehen, Erde im Mittelpunkt)



Jedoch ist ein Fehler in den Bildern!! Welcher wohl???

Die Umlaufgeschwindigkeit ist nicht konstant, sondern sie verhält sich gemäß Kepler:
- Die Quadrate der Umlaufzeiten von Planeten verhalten sich wie die Kuben der großen Halbachsen
- Ein Planet überstreicht in gleichen Zeiten gleiche Flächen.

Der Einfluß der Gravitation

Die Gravitationskraft F = gamma * Mm/r^2 zeigt die Abhängigkeit derselben vom Abstand des umlaufenden Körpers.

Schreiben wir dies in differentieller Form:

$x'' = -x/r^3$, $y'' = -y/r^3$.

unter Beachtung, daß:

$x^2 + y^2 = r^2$
$1/r^2 = \sqrt{(x'')^2+(y'')^2}$

Durch eine kleine Substitution wird die Angelegenheit vereinfacht:
 $u=x'$ and $v=y'$

Somit können 4 Gleichungen erster Ordnung niedergeschrieben werden:

```
> x:='x':  y:= 'y': u :='u': v:= 'v': t:= 't':
> ODEa := diff(x(t),t)=u(t):
ODEb := diff(y(t),t)=v(t):
ODEc := diff(u(t),t)=-x(t)/(x(t)^2+y(t)^2)^(3/2):
ODEd := diff(v(t),t)=-y(t)/(x(t)^2+y(t)^2)^(3/2):
```

```
eqns := [ODEa, ODEb, ODEc, ODEd];
```

$$eqns := \left[\frac{d}{dt}\,\mathrm{x}(t) = \mathrm{u}(t),\ \frac{d}{dt}\,\mathrm{y}(t) = \mathrm{v}(t),\ \frac{d}{dt}\,\mathrm{u}(t) = -\frac{\mathrm{x}(t)}{(\mathrm{x}(t)^2 + \mathrm{y}(t)^2)^{(3/2)}},\ \frac{d}{dt}\,\mathrm{v}(t) = -\frac{\mathrm{y}(t)}{(\mathrm{x}(t)^2 + \mathrm{y}(t)^2)^{(3/2)}}\right]$$

Differentialgleichungen werden durch Integration gelöst und dazu muß auch die Integrationskonstante bekannt sein. Dieses wird mit Hilfe der Anfangswertbedingung ermöglicht. Jedoch muß die Anfangswertbedingung vernünftig sein.

```
> ICx := x(0) = 1:
ICy := y(0) = 0:
ICu := u(0) = 0:
ICv := v(0) = 1:
solcurve := dsolve({ODEa, ODEb, ODEc, ODEd, ICx, ICy, ICu, ICv},
    [x(t), y(t), u(t), v(t)], numeric):
```

Jetzt schauen wir uns ausgewählte Ortspunkte an: Pi/2, Pi, 3Pi/2 und 2Pi:

```
> P1 := solcurve(Pi/2);
P2 := solcurve(Pi);
P3 := solcurve(3*Pi/2);
P4 := solcurve(2*Pi);
```

$$P1 := [t = 1.5707963267949,\ \mathrm{x}(t) = -0.487816757432923966\ \ 10^{-6},\ \mathrm{y}(t) = 0.99999924677821472,\ \mathrm{u}(t) = -1.00000028690815834,\ \mathrm{v}(t) = -0.129483408989805246\ \ 10^{-5}]$$

$$P2 := [t = 3.1415926535898,\ \mathrm{x}(t) = -0.999997747966278382,\ \mathrm{y}(t) = -0.390191275915263204\ \ 10^{-5},\ \mathrm{u}(t) = 0.479099051889369946\ \ 10^{-5},\ \mathrm{v}(t) = -1.00000129387785952]$$

$$P3 := [t = 4.7123889803847,\ \mathrm{x}(t) = 0.0000110542484625399638,\ \mathrm{y}(t) = -0.999996585358547740,\ \mathrm{u}(t) = 1.00000198868718027,\ \mathrm{v}(t) = 0.0000113844052061435122]$$

$$P4 := [t = 6.2831853071796,\ \mathrm{x}(t) = 0.999996166659835484,\ \mathrm{y}(t) = 0.0000198086268683804034,\ \mathrm{u}(t) = -0.0000200848149460668667,\ \mathrm{v}(t) = 1.00000191386782555]$$

Und dann lassen wir uns die Werte von x und y dieser Punkte geben:

```
> P1a := [op(2,P1[2]),op(2,P1[3])];
```

```
P2a := [op(2,P2[2]),op(2,P2[3])];
P3a := [op(2,P3[2]),op(2,P3[3])];
P4a := [op(2,P4[2]),op(2,P4[3])];
```

$P1a := [-0.487816757432923966\ 10^{-6}, 0.99999924677821472\ ]$

$P2a := [-0.999997747966278382\ , -0.390191275915263204\ \ 10^{-5}]$

$P3a := [0.0000110542484625399638\ , -0.999996585358547740\ ]$

$P4a := [0.999996166659835484\ , 0.0000198086268683804034\ ]$

Wir erhalten damit einen Kreis, periodisch in 2Pi. Das erstaunt nicht weiter.

```
> with(plots):
pointplot([seq([op(2,solcurve(t*.05)[2]),op(2,solcurve
(t*.05)[3])],
         t=0..125)], connect=true,title="Kreisbahn");
```



Wiederum das Spiel am Anfang gezeigt hier wiederholt: vom Kreis zur Ellipse. Zunächst jedoch machen wir ein Beispiel mit beliebigen Werden, um zu sehen ob wir eine Ellipse erhalten:

```
> e := 0.8; a := 1; tmax := 2*Pi*a*sqrt(a):
ICx := x(0) = a*(1-e);
ICy := y(0) = 0:
ICu := u(0) = 0:
ICv := v(0) = evalf(sqrt((1+e)/(a*(1-e))));
solcurve := dsolve({ODEa, ODEb, ODEc, ODEd, ICx, ICy,
ICu, ICv},
     [x(t), y(t), u(t), v(t)], numeric);
pointplot([seq([op(2,solcurve(t*tmax/100)[2]),
     op(2,solcurve(t*tmax/100)[3])],
         t=0..100)], connect=true,
scaling=constrained,title="Ellipsenbahn");
```

$e := 0.8$

$a := 1$

$ICx := x(0) = 0.2$

$ICv := v(0) = 3.000000000$

$solcurve$:= **proc** (x_rkf45) ... **end proc**



Kommen wir zurück auf unsere relativen Planetenbahnen und schauen uns die Erd-Merkur Bahn an.
Ich wechsel jetzt mal in der Radiusdarstellung auf astronomische Einheiten.

```
> a_erde := 1; tmax_erde := 2*Pi*a_erde*sqrt(a_erde);
ICxe := x(0) = a_erde*(1-e_erde);
ICye := y(0) = 0:
ICue := u(0) = 0:
ICve := v(0) = evalf(sqrt((1+e_erde)/(a_erde*(1-e_erde))));
solcurvee := dsolve({ODEa, ODEb, ODEc, ODEd, ICxe, ICye, ICue, ICve},
     [x(t), y(t), u(t), v(t)], numeric);
a_merk := r_merk/r_erde; tmax_merk := 2*Pi*a_merk*sqrt(a_merk);
ICxm := x(0) = a_merk*(1-e_merk);
ICym := y(0) = 0:
ICum := u(0) = 0:
ICvm := v(0) = evalf(sqrt((1+e_merk)/(a_merk*(1-e_merk))));
solcurvem := dsolve({ODEa, ODEb, ODEc, ODEd, ICxm, ICym, ICum, ICvm},
     [x(t), y(t), u(t), v(t)], numeric);
```

$a\_erde := 1$

$$tmax\_erde := 2\,\pi$$

$$ICxe := x(0) = 0.9833$$

$$ICve := v(0) = 1.016841803$$

$$solcurvee := \textbf{proc}\,(x\_rkf45)\ \ldots\ \textbf{end proc}$$

$$a\_merk := 0.3825096345$$

$$tmax\_merk := 0.4731442464\ \pi$$

$$ICxm := x(0) = 0.3038656536$$

$$ICvm := v(0) = 1.991869184$$

$$solcurvem := \textbf{proc}\,(x\_rkf45)\ \ldots\ \textbf{end proc}$$

```
> tmax := 2*max(tmax_erde, tmax_merk);
pointplot([seq(
[op(2,solcurvee(t*tmax/300)[2])-op(2,solcurvem
(t*tmax/300)[2]),
op(2,solcurvee(t*tmax/300)[3])-op(2,solcurvem(t*tmax/300)
[3])],
t=0..400)], connect=true,
scaling=constrained,title="Merkurbahn von der Erde aus
gesehen mit gravibedingter Geschwindigkeitskorrektur");
```

$$tmax := 4\,\pi$$



Schaut man intensiver hin, so erkennt man, daß einige der inneren Schleife kleiner sind als andere.

Versuchen wir das gleiche mal mit unserem Kometen Halley:

```
> ICxe := x(0) = a_erde*(1-e_erde);
ICye := y(0) = 0:
ICue := u(0) = 0:
ICve := v(0) = evalf(sqrt((1+e_erde)/(a_erde*(1-
e_erde))));
solcurve_erde := dsolve({ODEa, ODEb, ODEc, ODEd, ICxe,
ICye, ICue, ICve},
     [x(t), y(t), u(t), v(t)], numeric);
a_halley:=r_halley/r_erde;tmax_halley := evalf(2
*Pi*a_halley*sqrt(a_halley));
ICxh := x(0) = a_halley*(1-e_halley);
ICyh := y(0) = 0:
ICuh := u(0) = 0:
ICvh := v(0) = evalf(sqrt((1+e_halley)/(a_halley*(1-
e_halley))));
solcurve_halley := dsolve({ODEa, ODEb, ODEc, ODEd, ICxh,
ICyh, ICuh, ICvh},
     [x(t), y(t), u(t), v(t)], numeric);
tmax := 2*max(tmax_erde, tmax_halley);
pointplot([seq(
```

```
[op(2,solcurve_erde(t*tmax/1000)[2])-op(2,solcurve_halley
(t*tmax/1000)[2]),
op(2,solcurve_erde(t*tmax/1000)[3])-op(2,solcurve_halley
(t*tmax/1000)[3])],
t=-100..100)], connect=true,
scaling=constrained,title="Bahn des Kometen Halley mit
gravibedingter Geschwindigkeitskorrektur");
```

$$ICxe := x(0) = 0.9833$$

$$ICve := v(0) = 1.016841803$$

$$solcurve\_erde := \mathbf{proc}\,(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

$$a\_halley := 18.06420053$$

$$tmax\_halley := 482.4007641$$

$$ICxh := x(0) = 0.5419260159$$

$$ICvh := v(0) = 1.906615472$$

$$solcurve\_halley := \mathbf{proc}\,(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

$$tmax := 964.8015282$$



>
In Sonnennähe befindet sich halley innerhalb der Erdbahn. Die Bahngeschwindigkeit nimmt mit zunehmender Sonnennähe auch zu und
ist maximal groß im Perihel.

Machenb wir von der erdinneren Bahn mal eine Vergrößerung des Bildes

```
> pointplot([seq(
[op(2,solcurve_erde(t*tmax/5000)[2])-op(2,solcurve_halley
(t*tmax/5000)[2]),
op(2,solcurve_erde(t*tmax/5000)[3])-op(2,solcurve_halley
(t*tmax/5000)[3])],
t=-100..100)], connect=true,
scaling=constrained,title="Bahn des Kometen Halley
innerhalb der Erdbahn (Bahn 1)");
```



Diese merkwürdige Bahn stellt also die Spur des kometen Halley innerhalb der Erdbahn dar.
Das obige Bild zeigt die Situation, wenn Halley von der Erdseite aus in seine erdinnere Bahn eintaucht.

Das folgende Bild zeigt die Situation, wenn Halley von der erdabgewandten Seite in seine erdinnere Bahn eintaucht.
Hier sind die Kurven etwas weicher geformt.

```
> pointplot([seq(
[op(2,solcurve_erde(t*tmax/5000+Pi)[2])-op
(2,solcurve_halley(t*tmax/5000)[2]),
op(2,solcurve_erde(t*tmax/5000+Pi)[3])-op
(2,solcurve_halley(t*tmax/5000)[3])],
t=-100..100)], connect=true,
```

```
scaling=constrained,title="Bahn des Kometen Halley
innerhalb der Erdbahn (Bahn 2)");
```



>
>